

التطبيقات متعددة الطبقات

N-Tier Applications

نسخة غير منقحة

تحت التطوير

اعداد المهندس

محمد شاكر محمود وهيب البياتي

بغداد-2009

بسم الله الرحمن الرحيم

الحمد لله رب العالمين والصلاة والسلام على سيدنا محمد وعلى اله وصحبه الطيبين الطاهرين

اسئل الله سبحانه وتعالى ان يجعل هذا العمل خالصاً لوجهه الكريم

ارجو ممن يقرأ هذه الصفحات واعجبه ما كتبت ان يدعو لي ولأخي المرحوم عمر رحمه الله

اما بعد

مقدمة

يعد تعدد الطبقات في بناء التطبيقات شيء من الاشياء المهمة بالنسبة للمبرمج ، ولكي يبني المبرمج تطبيقاً رصينا متيناً فيجب عليه اتقان هذه التقنية المهمة ، والصراحة لم اجد احد من الاخوان يشرح هذه الطريقة بشكل جيد ، او ممكن يكون شرحها بشكل جيد ولكن لم يعطي مثال وافي ولذلك سوف نبقي في دوامة البحث في الانترنت وفي غير اماكن اسئل الله سبحانه وتعالى ان يوفقتي لاتمم عملي هذا بصورة جيدة ومفيدة ينتفع بها كل انسان يحب البرمجة ويرغب في تطوير امكانياته

اولاً يجب ان تعرف هذه التسميات

طبقة البيانات Data Tier or Data Layer

طبقة الاعمال Business Tier or Business Layer

طبقة العرض Presentation Tier or Presentation Layer

- الطبقة الاولى- طبقة البيانات - قاعدة البيانات وما تحويه من اجرائات مخزنة
- الطبقة الثانية- طبقة الاعمال- كل الـ Classes الخاصة بالاتصال بقاعدة البيانات والخاصة بالجداول
- الطبقة الثالثة – طبقة العرض – كل الـ Forms يمثل طبقة العرض


الخطوات الاساسية لبناء تطبيق متعدد الطبقات:-

1. بناء قاعدة البيانات
2. بناء Class خاص بالاتصال بقاعدة البيانات يحتوي على اعدادات الاتصال بقاعدة البيانات
3. بناء Class خاص بكل جدول موجود في قاعدة البيانات الخاصة بنا (طبعاً الجداول الغير ضرورية يمكن ان لا نعمل لها Class) كل Class من هذه الـ Classes يحتوي على كل الخصائص Properties والطرق Methods و الحقول Fields
4. تصميم طبقة العرض والتي تمثل الـ Form والتي سوف نتامل من خلالها بالطبقات الاخرى


الان باذن الله سوف نأخذ مثلاً بسيطاً لتعلم هذه التقنية

ابدا مشروع **WindowsApplication** جديد واتبع الخطوات التالية

الخطوة الاولى: بناء قاعدة البيانات سوف استخدم SQL Server Express 2005 لبناء قاعدة البيانات اتبع الصور




ثم اختر Service-based Database لتكوين قاعدة بيانات من نوع mdf كما في الشكل ادناه
*يمكن ان تختار اي نوع من انواع قواعد البيانات لكن نحن في مثالنا هذا سوف استخدم النوع mdf



بعد ان انشئة قاعدة البيانات قم بسحبها الى المسار التالي

Bin\Debug كما في الصورة ادناه



اسم قاعدة البيانات DataBase1.mdf تحتوي على جدولين هما

tblCity و tblUsers

قم ببناء الجدولين وضبط الخصائص كما مبين في الصور ادناه

The screenshot displays the SQL Server Enterprise Manager interface. On the left, the Server Explorer shows the 'tblUsers' table selected under 'Database1.mdf'. The main window shows the 'Table Designer' for 'tblUsers'. The columns are listed as follows:

Column Name	Data Type	Allow Nulls
userID	int	<input type="checkbox"/>
username	nvarchar(200)	<input checked="" type="checkbox"/>
city	int	<input checked="" type="checkbox"/>

The 'Column Properties' pane for 'userID' is expanded, showing the following settings:


Property	Value
Collation	<data>
Computed Column Specification	
Condensed Data Type	int
Description	
Deterministic	Yes
DTS-published	No
Full-text Specification	No
Has Non-SQL Server Subscriber	No
Identity Specification	Yes
(Is Identity)	Yes
Identity Increment	1
Identity Seed	1
Indexable	Yes
Merge-published	No
Not For Replication	No
Replicated	No


The screenshot displays the SQL Server Enterprise Manager interface. On the left, the 'Server Explorer' shows a tree view of the database structure, with 'tblcity' selected under the 'Tables' folder. The main pane shows the 'dbo.tblcity: T...DATABASE1.MDF' table structure. The 'cityID' column is highlighted with a red box, showing it is of type 'int' and 'Allow Nulls' is unchecked. Below the table structure, the 'Column Properties' window is open for 'cityID', with the 'Identity Specification' section highlighted by a red box. This section shows 'Identity Specification' set to 'Yes', '(Is Identity)' set to 'Yes', 'Identity Increment' set to '1', and 'Identity Seed' set to '1'.

Column Name	Data Type	Allow Nulls
cityID	int	<input type="checkbox"/>
city	nvarchar(50)	<input checked="" type="checkbox"/>

Property	Value
Collation	<database c
Computed Column Specification	
Condensed Data Type	int
Description	
Deterministic	Yes
DTS-published	No
Full-text Specification	No
Has Non-SQL Server Subscriber	No
Identity Specification	Yes
(Is Identity)	Yes
Identity Increment	1
Identity Seed	1
Indexable	Yes
Merge-published	No
Not For Replication	No
Replicated	No

الخطوة الثانية: اضع مشروعاً جديداً الى مشروعنا الحالي من نوع Class Library وقم بتسميته اي اسم ترغب به انا بالنسبة الي سوف اسميه MuhammadData كما في الصور ادناه





الآن نقوم الآن بتكوين الـ Classes لكل جدول كما ذكرنا سابقاً بالاضافة الى Class خاص للاتصال بقاعدة البيانات

اضف خمس Classes وقم بتسميتها كما يلي

clsDBconnection


clsUsers

clsCMDUsers

clsCity

clsCMDCity

طبعاً الـ Classes اعلاه تضاف الى مشروع الـ Class Library كما في الصورة ادناه



ثم ما عليك الا ان تدخل الاسم مثل clsCity او اكمل الباقي..


الخطوة الثالثة : اعداد الاتصال بقاعدة البيانات : هذه الخطوة تعتبر من الخطوات المهمة التي يمكن وصفها بقلب البرنامج. لا يمكن ان تضع سلسلة الاتصال داخل احد الاصناف Class لان هذا الصنف سوف يكون ملف DLL كما سنرى , وكما معروف ملف الـ DLL لا يمكن تحريره (للمستخدم العادي) لذلك عند وضع برنامجك على هيئة ملف Setup وتنصيبه على حواسيب المستخدمين كيف ستعدل عندئذ سلسلة الاتصال هل سوف تحدد المستخدم بمسار لوضع قاعدة بياناته ام ماذا ؟؟؟؟

الحل هو: ان تستخدم ملف الاعدادات app.config ولكن يجب ان تنتبه دوما نحن نملك الان برنامجين برنامج من نوع WindowsApplication ونوع اخر هو Class Library الان ايهما سوف تستخدم ملف اعدادته هذا ام هذا ؟؟؟


كيف؟؟

اولاً اضبط الاعدادات في البرنامج الذي نوعه Class Library قم بالنقر المزدوج على My Project وفي الخاصية Name اكتب conString


ويجب ان يكون النوع **Connection String** بعدها اضغط الزر لجلب سلسلة الوصول لقاعدة البيانات التي انشئناها في البداية لاحظ الصور ادناه لتكتمل عندك الصورة



لجلب سلسلة الاتصال



افتح ملف الاعدادات **app.config** وقم بازالة علامات الاقتباس المحيطة بالمسار **"C:\Documents and Settings\...\\"** وكن حذراً وقم بنسخ الكود الخاص بسلسلة الاتصال كما يلي



قم الان بالنقر المزدوج على My Project الخاص بـ WindowsApplication واذهب الى Settings وضع اسماً وليكن conString ونوعه Class Library كما فعلنا سابقاً في المشروع Class Library

بعدها افتح app.config واستبدل سلسلة الاتصال الخاصة بالـ WindowsApplication بالسلسلة الخاصة ببرنامج الـ Class Library وحفظ التغييرات قم بحذف الـ app.config الخاص بـ مشروع Class Library

الان تعتبر مشكلة الاتصال بقاعدة بياناتنا منتهية اتمنى يكون شرحي واضح في هذه النقطة

نعود لكتابة الاكواد الخاصة بالتعامل مع سلسلة الاتصال بقاعدة البيانات

الكود النهائي للـ clsDBconnection

```
Imports System.Data.SqlClient
Public Class clsDBConnection


    Public Shared Function GetConnection() As SqlConnection
        Dim connectionString As String = My.Settings.conString
        Return New SqlConnection(connectionString)
    End Function

End Class
```


قبل كتابة بقية الاكواد سوف نتطرق الى موضوع استخدام الـ ClassDiagram بشكل سريع جداً


----- هذه المنطقة خاصة بالـ ClassDiagram يمكنك تجاوزها هي بس للتذكير حتى تسهل عليك كتابت الحقول والخصائص -----

الان سوف استخدم ClassDiagram وهو شي خيارى تستطيع الاستغنى عنه. بالحقية الـ ClassDiagram سوف يساعدنا في كتابة الكود الخاص بالحقول والخصائص قليلا وهو جميل جداً (لاتخف هو ليس Wizard بمعنى الكلمة) اتبع الصور التالية لكي تتضح الصورة امامك في كيفية استخدامه



سوف تظهر الشاشة ادناه





سوف تظهر الصورة ادناه لكنها فارغة اي ليس بها خصائص ولا حقول اذا استطعت اكتب الخصائص والحقول التالية

Properties			
City	String		Public
UserID	Integer		Public
UserName	String		Public
<add property>			
Fields			
_City	String		Private
_UserID	Integer		Private
_userName	String		Private
<add field>			
Events			
<add event>			

*بعد كتابتك للخصائص والحقول لم تنتهي مهمتك بعد فعليك اضافة بعض الكود الى الـ Class

----- انتهت المنطقة الخاصة بالـ ClassDiagram -----

```
Public Class clsUsers

#Region " Fields تعريف الحقول ويجب مراعاة النوع "

    Private _UserID As Integer
    Private _userName As String
    Private _City As String ' رقمي بالاصل انه مع سترنك نصي النوع وضعت اني تلاحظ لو
    ' string نوعه ولكن نفسه الحقل على يحتوي اخر بجدول مربوط الجدول لانه وذلك
    ' المدينة اسم لظهار inner join استخدم وسوف

#End Region

#Region " constructor تعريف الباني تعريف "
    Public Sub New()

        End Sub
#End Region

#Region "properties خصائصه يقابله الجدول في عمود لكل الخصائص "

    Public Property UserID() As Integer
        Get
            Return _UserID
        End Get
        Set(ByVal value As Integer)
            _UserID = value
        End Set
    End Property

    Public Property UserName() As String
        Get
            Return _userName
        End Get
        Set(ByVal value As String)
            _userName = value
        End Set
    End Property

    Public Property City() As String
        Get
            Return _City
        End Get
        Set(ByVal value As String)
            _City = value
        End Set
    End Property
#End Region

#End Region

End Class
```

```

Imports System.Data.SqlClient

Public Class clsCMDUsers

#Region " AddNew - UpDate -Saerch Methods 'البحث والتعديل والحذف بالاضافة الخاصة الطرق'"
#Region " AddNew الاضافة "

    Public Shared Function AddUser(ByVal User As clsUsers) As Integer
        Dim connection As SqlConnection = clsDBConnection.GetConnection
        Dim insertStatement As String _
            = "INSERT tblUsers (username,city) " _
            & "VALUES (@username,@city)"
        Dim insertCommand As New SqlCommand(insertStatement, connection)
        insertCommand.Parameters.AddWithValue("@username", User.UserName)
        insertCommand.Parameters.AddWithValue("@city", User.City)

        Try
            connection.Open()
            insertCommand.ExecuteNonQuery()
            Dim selectStatement As String _
                = "SELECT IDENT_CURRENT('tblUsers') FROM tblUsers"
            Dim selectCommand As New SqlCommand(selectStatement, connection)
            Dim userID As Integer = CInt(selectCommand.ExecuteScalar)
            Return userID
        Catch ex As SqlException
            Throw ex
        Finally
            connection.Close()
        End Try
    End Function

#End Region

#Region " Update التحديث "
    Public Shared Function UpdateUser(ByVal oldUser As clsUsers, ByVal newUser As clsUsers) As Boolean
        Dim connection As SqlConnection = clsDBConnection.GetConnection()
        Dim updateStatement As String _
            = "UPDATE tblUsers SET " _
            & "UserName = @NewUserName, " _
            & "City = @NewCity " _
            & "WHERE userID = @OlduserID "
        Dim updateCommand As New SqlCommand(updateStatement, connection)
        updateCommand.Parameters.AddWithValue("@NewUserName", newUser.UserName)
        updateCommand.Parameters.AddWithValue("@NewCity", newUser.City)

        updateCommand.Parameters.AddWithValue("@OlduserID", oldUser.UserID)
        updateCommand.Parameters.AddWithValue("@Oldusername", oldUser.UserName)
        updateCommand.Parameters.AddWithValue("@OldCity", oldUser.City)

        Try
            connection.Open()
            Dim count As Integer = updateCommand.ExecuteNonQuery()

```



```

        If count > 0 Then
            Return True
        Else
            Return False
        End If
    Catch ex As SqlException
        Throw ex
    Finally
        connection.Close()
    End Try
End Function

#End Region

#Region "Search كاملاً الاسم او الاسم من جزء تمرير بواسطة البحث"

Public Shared Function GetUserName(ByVal mykeyWord As String) As List(Of clsUsers)
    Dim userList As List(Of clsUsers) = New List(Of clsUsers)
    Dim connection As SqlConnection = clsDBConnection.GetConnection
    Dim selectStatement As String _
        = "SELECT userid,username,tblcity.city FROM tblusers inner JOIN tblcity on
tblusers.city = tblcity.cityID WHERE userName LIKE '%' + @mykeyWord + '%"
    Dim selectCommand As SqlCommand = New SqlCommand(selectStatement, connection)
    selectCommand.Parameters.AddWithValue("@mykeyWord", mykeyWord)
    connection.Open()
    Dim reader As SqlDataReader =
selectCommand.ExecuteReader(CommandBehavior.CloseConnection)
    Dim users As clsUsers
    Do While reader.Read
        users = New clsUsers
        users.UserID = reader("userid").ToString ' 0 userid
        users.UserName = reader("userName").ToString '1 username
        users.City = reader("City").ToString '2 city
        userList.Add(users)
    Loop
    reader.Close()
    ' بالنتائج قائمة ارجاع او بتصدير يقوم
    Return userList

End Function
#End Region

#Region " Just Example to use this method to return DataSet ادناه الكود استخدام يمكنك
داتاسيت لارجاع كمثال "
' DataSet تعيد الطريقة هذه
'Public Shared Function GetUserName(ByVal mykeyWord As String) As DataSet
'    Dim DS1 As New DataSet
'    Dim connection As SqlConnection = clsDBConnection.GetConnection
'    Dim selectStatement As String = "SELECT userid,username as [الاسم],tblcity.city
as [المدينة] FROM tblusers inner JOIN tblcity on tblusers.city = tblcity.cityID WHERE
userName LIKE '%' + @mykeyWord + '%"
'    Dim selectCommand As SqlCommand = New SqlCommand(selectStatement, connection)
'    selectCommand.Parameters.AddWithValue("@mykeyWord", mykeyWord)

'    connection.Open()
'    Dim DataAdapter1 As New SqlDataAdapter(selectCommand)
'    DS1.Clear()
'    DataAdapter1.Fill(DS1, "tblusers")
'    connection.Close()

```

```

    '    Return DS1
    'End Function

#End Region

#Region " 'I use this function when I click on DtatGridView at form1. على النقر عند
النمى مربعات فى البيانات تعرض الداتاكرى
Public Shared Function GetByUserID(ByVal userID As Integer) As clsUsers
    Dim connection As SqlConnection = clsDBConnection.GetConnection
    Dim selectStatement As String = "SELECT userid,username,tblcity.city FROM
tblusers inner JOIN tblcity on tblusers.city = tblcity.cityID WHERE userID = @userID"
    Dim selectCommand As SqlCommand = New SqlCommand(selectStatement, connection)
    selectCommand.Parameters.AddWithValue("@userID", userID)
    connection.Open()
    Dim reader As SqlDataReader _
        = selectCommand.ExecuteReader(CommandBehavior.SingleRow) ' سطر جلب واحد
    Dim myUsers As clsUsers
    If reader.Read Then
        myUsers = New clsUsers
        myUsers.UserID = reader("userID").ToString
        myUsers.UserName = reader(1).ToString
        myUsers.City = reader(2).ToString
    Else
        myUsers = Nothing
    End If
    reader.Close()
    Return myUsers
End Function
#End Region

#Region "Delete User مستخدم لحذف"
Public Shared Function DeleteUser(ByVal User As clsUsers) As Boolean
    Dim connection As SqlConnection = clsDBConnection.GetConnection
    Dim deleteStatement As String _
        = "DELETE FROM tblUsers " _
        & "WHERE userID = @userID " _
        & "and username = @username "

    Dim deleteCommand As New SqlCommand(deleteStatement, connection)
    deleteCommand.Parameters.AddWithValue("@userID", User.UserID)
    deleteCommand.Parameters.AddWithValue("@username", User.UserName)
    deleteCommand.Parameters.AddWithValue("@city", User.City)
    Try
        connection.Open()
        Dim count As Integer = deleteCommand.ExecuteNonQuery
        If count > 0 Then
            Return True
        Else
            Return False
        End If
    Catch ex As SqlException
        Throw ex
    Finally
        connection.Close()
    End Try

```

```
End Function
#End Region

#End Region
End Class
```

الكود النهائي للـ clsCity

```
Imports System.Data.SqlClient

Public Class clsCity

    Private _CityID As Integer
    Private _City As String

    Public Sub New()

    End Sub

    Public Property CityID() As Integer
        Get
            Return _CityID
        End Get
        Set(ByVal value As Integer)
            _CityID = value

        End Set
    End Property

    Public Property City() As String
        Get
            Return _City
        End Get
        Set(ByVal value As String)
            _City = value

        End Set
    End Property

End Class
```

الكود النهائي للـ clsCMDCity

```
Imports System.Data.SqlClient

Public Class clsCMDCity

    Public Shared Function GetCityList() As List(Of clsCity)
        Dim cityList As New List(Of clsCity)
        Dim connection As SqlConnection = clsDBConnection.GetConnection
        Dim selectStatement As String _
            = "SELECT * FROM tblcity" _
        Dim selectCommand As New SqlCommand(selectStatement, connection)
        Try
            connection.Open()
            Dim reader As SqlDataReader = selectCommand.ExecuteReader
            Dim city As clsCity
            Do While reader.Read
                city = New clsCity
                city.CityID = CInt(reader("Cityid"))
                city.City = reader("City").ToString
                cityList.Add(city)
            Loop
            reader.Close()
        Catch ex As SqlException
            Throw ex
        Finally
            connection.Close()
        End Try
        Return cityList
    End Function

    Public Shared Function Addcity(ByVal city As clsCity) As Integer
        Dim connection As SqlConnection = clsDBConnection.GetConnection
        Dim insertStatement As String _
            = "INSERT tblcity (city)" _
            & "VALUES (@city)"
        Dim insertCommand As New SqlCommand(insertStatement, connection)

        insertCommand.Parameters.AddWithValue("@city", city.City)

        Try
            connection.Open()
            insertCommand.ExecuteNonQuery()
            Dim selectStatement As String _
                = "SELECT IDENT_CURRENT('tblcity') FROM tblcity"
            Dim selectCommand As New SqlCommand(selectStatement, connection)
            Dim RevID As Integer = CInt(selectCommand.ExecuteScalar)
            Return RevID
        Catch ex As SqlException
            Throw ex
        Finally
            connection.Close()
        End Try
    End Function
End Class
```

```

Public Shared Function Updatecity(ByVal oldcity As clsCity, _
    ByVal newcity As clsCity) As Boolean
    Dim connection As SqlConnection = clsDBConnection.GetConnection()
    Dim updateStatement As String _
        = "UPDATE tblcity SET " _
        & "city=@newcity " _
        & "WHERE cityID = @OldcityID " _
        & "And city = @oldcity "
    Dim updateCommand As New SqlCommand(updateStatement, connection)
    updateCommand.Parameters.AddWithValue("@newcity", newcity.City)

    updateCommand.Parameters.AddWithValue("@oldcityID", oldcity.CityID)
    updateCommand.Parameters.AddWithValue("@oldcity", oldcity.City)

    Try
        connection.Open()
        Dim count As Integer = updateCommand.ExecuteNonQuery
        If count > 0 Then
            Return True
        Else
            Return False
        End If
    Catch ex As SqlException
        Throw ex
    Finally
        connection.Close()
    End Try
End Function

Public Shared Function GetcityBycityID(ByVal cityID As Integer) As clsCity
    Dim city As New clsCity
    Dim connection As SqlConnection = clsDBConnection.GetConnection
    Dim selectStatement As String _
        = "SELECT * FROM tblcity WHERE cityID = @cityID"
    Dim selectCommand As New SqlCommand(selectStatement, connection)
    selectCommand.Parameters.AddWithValue("@cityID", cityID)
    Try
        connection.Open()
        Dim reader As SqlDataReader _
            = selectCommand.ExecuteReader(CommandBehavior.SingleRow)
        If reader.Read Then

            city.CityID = CInt(reader("cityID"))
            city.City = reader("city").ToString

        Else
            city = Nothing
        End If
        reader.Close()
    Catch ex As SqlException
        Throw ex
    Finally
        connection.Close()
    End Try
    Return city
End Function

Public Shared Function Deletecity(ByVal city As clsCity) As Boolean
    Dim connection As SqlConnection = clsDBConnection.GetConnection

```

```


Dim deleteStatement As String _
    = "DELETE FROM tblcity " _
    & "WHERE cityID = @cityID " _
    & "and city = @city "

Dim deleteCommand As New SqlCommand(deleteStatement, connection)
deleteCommand.Parameters.AddWithValue("@cityID", city.CityID)
deleteCommand.Parameters.AddWithValue("@city", city.City)


Try
    connection.Open()
    Dim count As Integer = deleteCommand.ExecuteNonQuery()
    If count > 0 Then
        Return True
    Else
        Return False
    End If
Catch ex As SqlException
    Throw ex
Finally
    connection.Close()
End Try
End Function
End Class

```

بعد الانتهاء من كتابة الاكواد يجب عليك عمل Build للـ Class Library بالنقر يمين ثم Build هذه العملية سوف تنتج ملف واحد اسمه MuhammadData.dll بعد حصولنا على هذا الملف يجب اضافته كمرجع الى الـ WindowsApplication وذلك بالنقر يمين على WindowsApplication ثم اختر Add Reference ومن صفحة التثبيت المسماة Project اختر MuhammadData ثم OK



الآن نصمم الواجهة او طبقة العرض ضع الادوات الملائمة و سمها كما مبين ادناه

الواجهة الرئيسية

طبعاً لا تنسى حبيبة القلب الداتا كورد DataGridView1

الكود النهائي للـ Form1

```
Imports MuhammadData 'Classes استدعاء اسم فضاء المشروع الذي يحتوي الـ
Public Class Form1
    Dim users As clsUsers ' الكلاس من نسخة انشاء
    Private usersList As List(Of clsUsers) ' الكلاس من العائدة القوائم استخدام

    Private Sub Form1_Load(ByVal sender As System.Object, ByVal e As System.EventArgs)
Handles MyBase.Load
        LoadCOMBOboxes() ' بالقيم الـ لتحميل
    End Sub

    Private Sub LoadCOMBOboxes() '
        ' بالقيم cmbCity الـ لتحميل
        Dim CityList As List(Of clsCity)
        CityList = clsCMDCity.GetCityList
        cmbCity.DataSource = CityList
        cmbCity.DisplayMember = "City"
        cmbCity.ValueMember = "Cityid"
        cmbCity.SelectedIndex = -1
    End Sub
End Class
```

```

End Sub

Private Sub PutUserData(ByVal users As clsUsers)
    'الكلاس الى النص مربع من القيم لحشر'
    users.UserName = txtUserName.Text
    users.City = CInt(cmbCity.SelectedIndex + 1)

End Sub

Private Sub btnSave_Click(ByVal sender As Object, ByVal e As System.EventArgs)
Handles btnSave.Click

    'حفظ او جديدة قيمة ادخال يريد المستخدم هل لمعرفة'
    'التعديل'
    If txtUserID.Text = "" Then
        If Trim(txtUserName.Text) = "" Or cmbCity.SelectedIndex = -1 Then
            MsgBox("الادخال عملية اتم ")
            Exit Sub
        End If
        users = New clsUsers
        Me.PutUserData(users) 'الجديدة القيم جلب'
        Try
            users.UserID = clsCMDUsers.AddUser(users)
            MessageBox.Show("البيانات حفظ تم لقد")
        Catch ex As Exception
            MessageBox.Show(ex.Message, ex.GetType.ToString)
        End Try
    Else
        If Trim(txtUserName.Text) = "" Or cmbCity.SelectedIndex = -1 Then
            MsgBox("الادخال عملية اتم ")
            Exit Sub
        End If
        Dim newuser As New clsUsers
        newuser.UserID = users.UserID
        Me.PutUserData(newuser)
        Try
            If Not clsCMDUsers.UpdateUser(users, newuser) Then
                MessageBox.Show("خطا", "التحديث عملية اثنا خطأ يوجد , الحفظ يتم لم ")
                Me.DialogResult = DialogResult.Retry
            Else
                users = newuser
                MsgBox("السجل على التعديلات حفظ تم لقد ")
            End If
        Catch ex As Exception
            MessageBox.Show(ex.Message, ex.GetType.ToString)
        End Try
    End If
    clsDBConnection.GetConnection.Close()

    ClearAllBoxes()

    'الداتا في القيم عرض لاعادة'
    'كرد'
    usersList = clsCMDUsers.GetUserName(txtSearch.Text)
    DataGridView1.DataSource = usersList

End Sub
'الحقول لتنظيف'
Private Sub ClearAllBoxes()

```



```

        txtUserID.Text = ""
        txtUserName.Text = ""
        cmbCity.SelectedIndex = -1
    End Sub
    'النصوص مربعات في البيانات تعرض كرد الداتا على الضغط عند'
    'المستخدم رقم على اعتماداً'
    Private Sub GetUser(ByVal userID As Integer)
        Try
            users = clsCMDUsers.GetByUserID(userID)
            If users Is Nothing Then
                MessageBox.Show("شي لا يوجد !!!!!")
            Else
                Me.DisplayUser()

            End If
        Catch ex As Exception
            MessageBox.Show(ex.Message, ex.GetType.ToString)
        End Try

    End Sub
    'النصوص مربعات في القيم لعرض'
    Private Sub DisplayUser()
        ClearAllBoxes()
        txtUserID.Text = users.UserID.ToString
        txtUserName.Text = users.UserName
        cmbCity.Text = users.City.ToString
    End Sub

    Private Sub btnDelete_Click(ByVal sender As System.Object, ByVal e As
System.EventArgs) Handles btnDelete.Click
        Try
            Dim result As DialogResult _
                = MessageBox.Show("&
users.UserID & "?", _
                                "الحذف تأكيد", MessageBoxButtons.YesNo, _
                                MessageBoxIcon.Question)
            If result = DialogResult.Yes Then
                Try
                    If Not clsCMDUsers.DeleteUser(users) Then
                        MessageBox.Show("مشكلة هناك ")
                        GetUser(users.UserID)
                        If users IsNot Nothing Then
                            Me.DisplayUser()
                        Else
                            Me.ClearAllBoxes()
                        End If
                    Else
                        Me.ClearAllBoxes()
                    End If
                Catch ex As Exception
                    MessageBox.Show(ex.Message, ex.GetType.ToString)
                End Try
            End If
        Catch ex As Exception
            MsgBox(ex.Message.ToString)
        End Try
        ClearAllBoxes()
        usersList = clsCMDUsers.GetUserName(txtSearch.Text)
    End Sub

```

```

        DataGridView1.DataSource = usersList
    End Sub

    Private Sub btnAddNew_Click(ByVal sender As System.Object, ByVal e As
System.EventArgs) Handles btnAddNew.Click
        ClearAllBoxes()
    End Sub

    Private Sub btnSearch_Click(ByVal sender As Object, ByVal e As System.EventArgs)
Handles btnSearch.Click

        usersList = clsCMDUsers.GetUserName(txtSearch.Text) ' باستخدام اسم عن للبحث
الدالة getUsername
        DataGridView1.DataSource = usersList ' الداتاكرد في العائدة النتائج عرض'

    End Sub

    Private Sub DataGridView1_CellClick(ByVal sender As Object, ByVal e As
System.Windows.Forms.DataGridViewCellEventArgs) Handles DataGridView1.CellClick
        ' الدالة الى واسناده المستخدم رقم جلب'
        Try
            Dim i As Integer = e.RowIndex
            Dim row As DataGridViewRow = DataGridView1.Rows(i)
            Dim myuser As clsUsers = CType(row.DataBoundItem, clsUsers)


            Dim userID As Integer = myuser.UserID
            Me.GetUser(userID)
        Catch ex As Exception
            MsgBox(ex.Message.ToString)
        End Try
    End Sub

    Private Sub btnShowCityForm_Click(ByVal sender As System.Object, ByVal e As
System.EventArgs) Handles btnShowCityForm.Click
        ' المدينة نافذة لعرض'
        Dim frmcity As New frmcity
        frmcity.ShowDialog()
    End Sub

    Private Sub cmbCity_Click(ByVal sender As Object, ByVal e As System.EventArgs)
Handles cmbCity.Click
        LoadCOMBOboxes() ' المنسدلة القائمة الى القيم لتحميل'
    End Sub
End Class

```

الان النافذة الخاصة بالمحافظات frmcity



الكود النهائي للـ frmCity

```
Imports MuhammadData

Public Class frmcity

    Dim city As clsCity

    Private Sub frmcity_Load(ByVal sender As System.Object, ByVal e As System.EventArgs)
Handles MyBase.Load
        LoadDataToGrid()

    End Sub
    Private Sub PutcityData(ByVal city As clsCity)

        city.City = txtCity.Text

    End Sub

    Private Sub btnSave_Click(ByVal sender As Object, ByVal e As System.EventArgs)
Handles btnSave.Click
        If txtCityID.Text = "" Then
            city = New clsCity
            Me.PutcityData(city)
            Try
                city.CityID = clsCMDCity.Addcity(city)
                MessageBox.Show("البيانات حفظ تم لقد")
            Catch ex As Exception
                MessageBox.Show(ex.Message, ex.GetType.ToString)
            End Try
        Else
```

```

Dim newcity As New clsCity
newcity.CityID = city.CityID
Me.PutcityData(newcity)
Try
    If Not clsCMDCity.Updatecity(city, newcity) Then
        MessageBox.Show("خطا", "التحديث عملية ائنا خطأ يوجد , الحفظ يتم لم ")
        Me.DialogResult = DialogResult.Retry
    Else
        city = newcity
        MsgBox(" السجل على التعديلات حفظ تم لقد ")
    End If
Catch ex As Exception
    MessageBox.Show(ex.Message, ex.GetType.ToString)
End Try

End If
clsDBConnection.GetConnection.Close()

ClearAllBoxes()
LoadDataToGrid()

End Sub

Private Sub ClearAllBoxes()
    txtCity.Text = ""
    txtCityID.Text = ""
End Sub

Private Sub LoadDataToGrid()
    Dim cityList As List(Of clsCity)
    cityList = clsCMDCity.GetCityList
    DataGridView1.DataSource = cityList
End Sub

Private Sub Getcity(ByVal cityID As Integer)
    Try
        city = clsCMDCity.GetcityBycityID(cityID)
        If city Is Nothing Then
            MessageBox.Show("!!!!!! شي لا يوجد ")
        Else
            Me.Displaycity()
        End If
    Catch ex As Exception
        MessageBox.Show(ex.Message, ex.GetType.ToString)
    End Try
End Sub

Private Sub Displaycity()
    txtCityID.Text = city.CityID
    txtCity.Text = city.City
End Sub

Private Sub btnDelete_Click(ByVal sender As System.Object, ByVal e As
System.EventArgs) Handles btnDelete.Click
    Dim result As DialogResult

```

```

        = MessageBox.Show("رقم السجل حذف تريد هل" & city.CityID & "?", _
        "الحذف تأكيد", MessageBoxButtons.YesNo, _
        MessageBoxIcon.Question)
    If result = DialogResult.Yes Then
        Try
            If Not clsCMDCity.Deletecity(city) Then
                MessageBox.Show("مشكلة هناك ")
                Getcity(city.CityID)
                If city IsNot Nothing Then
                    Me.Displaycity()
                Else
                    Me.ClearAllBoxes()
                End If
            Else
                Me.ClearAllBoxes()
            End If
        Catch ex As Exception
            MessageBox.Show(ex.Message, ex.GetType.ToString)
        End Try
    End If
    DataGridView1.DataSource = Nothing
    LoadDataToGrid()
End Sub

Private Sub btnClearBoxes_Click(ByVal sender As Object, ByVal e As System.EventArgs)
Handles btnAddNew.Click
    ClearAllBoxes()
End Sub

Private Sub btnShowAll_Click(ByVal sender As System.Object, ByVal e As
System.EventArgs) Handles btnShowAll.Click
    LoadDataToGrid()
End Sub


Private Sub DataGridView1_RowEnter(ByVal sender As Object, ByVal e As
System.Windows.Forms.DataGridViewCellEventArgs) Handles DataGridView1.RowEnter
    Try
        Dim i As Integer = e.RowIndex
        Dim row As DataGridViewRow = DataGridView1.Rows(i)
        Dim city As clsCity = CType(row.DataBoundItem, clsCity)

        'MessageBox.Show(expen.ExpID)
        Dim cityID As Integer = city.CityID
        Me.Getcity(cityID)
    Catch ex As Exception
        Exit Sub
    End Try
End Sub
End Class


```

اضف مشروعاً ثالثاً ينضم الى المشروعين السابقين يكون نوعه Project Setup


File>Add>New Project




لاضافة محتويات المشروع من اصناف و الى اخره من موجودات اخرى في البرنامج انظر الصورة التالية




بعد النقر على Project Output سوف تظهر النافذة ادناه اختر من القائمة المنسدلة المشروع الرئيسي وهو WindowsApplication1 او اسم مشروعك






بعدها يجب ان نصل الى قاعدة البيانات المخزونه في المسار

C:\Documents and Settings\Muhammad\My Documents\Visual Studio 2008\Projects\WindowsApplication1\bin\Debug



بعد ان قمنا باضافة قاعدة البيانات الى مشروع التنصيب نقوم باضافة ايقون حتى يمكن وضعه على سطح المكتب او قائمة ابدأ

انقر يمين على Primary output from windowsApplication1(Active) واختر انشاء مختصر




بعد انشاء المختصر سم المختصر كما يحلو لك انا سميتة Muhammad App 2009 حلو مووو

قم بنقل هذا المختصر الى User's Program menu اسحبه وافلته هذا يعني انك بعد تنصيبك للبرنامج سوف ترى المختصر موجود في قائمة ابدأ


قم بانشاء مختصر اخر كما فعلنا توأ وضعه في User's Desktop

الان اضع ايقون مميز لتميز برنامجك انقر يمين ثم Add>File




بعدها ابحث عن ايقونه جميله وضعها

انقر على User's Program menu واختر الايقون وضبط الخاصية كما في الصورة ادناه



إذا اردت ان تضع ايقونه او مختصر الى قائمة Add/Remove Programs لاحظ الصورة ادناه




ثم اختر الايقونه كما اخترناها من قبل


الحقيقة الخيارات كثيرة ولكن ندعها لك لتكتشف المزيد

هل ترغب باضافة SQL server الى مشروعك حتى يتمكن برنامجك من الاشتغال على اي حاسبة اخرى؟؟

اكيد ترغب بذلك قم بالنقر يمين على Setup>Properties سوف تظهر النافذة التالية




بعد النقر على زر Prerequisites قم باختيار البرامج التي تعجبك والتي يتاثر برنامج بها في حال لم تكن موجودة عند المستخدم او الزبون




الآن الخطوة الأخيرة لإتمام المشروع هي عمل Build للمشروع Setup انقر يمين على Setup ثم Build

اذهب إلى مسار المشروع Setup1\Debug سوف تجد كل الملفات المطلوبة


بعدها قم بتنصيب البرنامج وشغل البرنامج إذا لم تم الاتصال بقاعدة البيانات إذا إلى مسار التنصيب كما في الصورة ادناه وقم بتحرير ملف الإعدادات



غير المسار فقط وحفظ التغييرات



بعدها قم بتشغيل البرنامج وشوف النتيجة



ادعو الله سبحانه وتعالى ان اكون وفقت في عملي هذا واعتذر عن الاخطاء والسهو

اسئل الله سبحانه وتعالى ان يدخل اخي عمر رحمه الله الجنة وان يزوجه من الحور العين

امين

مع تحياتي

Eng_muhammad76